ENTERPRISE
COMPUTERS LTD

31-37 Hoxton Street
London N1 6NJ
Telephone 01-739 4282
Telex 22717 ENTER G

Application note number 20

Sorting strings in IS-BASIC.

One of the most common tasks a computer must carry out is the sorting of strings into alphabetical order. There are various methods of doing this but the one used in this application is known as a bubble sort. The name comes from the way that the 'lowest value' words appear to rise to the top during the sorting process.

The program compares two strings in a list: if the 'lowest value' string of the two is in the highest place the two strings are swapped. This type of sorting is faster than if two adjacent strings were tested because unnecessary swaps are reduced.

The procedure SORT is the actual sorting part of the program and can be used on its own. However it is a good idea to type the whole program in so that you can see how it works.

Parts of the listing such as 'TEXT 80' etc. are not essential to the running of the program but they give a neater display.

```
100 PROGRAM "BUBBLE_SORT"
110 ! Written by Gerald R Morgan.
120 TEXT 80
130 SET £102:PALETTE 7,0,121,121
140 PRINT "How many words to you want to sort (1 to 15)?"
150 INPUT PROMPT ">":AMOUNT
160 IF AMOUNT<2 OR AMOUNT>15 THEN 150
170 DIM ARRAY$(AMOUNT+2)
180 PRINT :PRINT :PRINT "Ok. Enter your" AMOUNT "words (max 14 letters):"
190 FOR X=1 TO AMOUNT
200   PRINT X;
210   INPUT PROMPT ">":ARRAY$(X)
220   IF LEN(ARRAY$(X))>14 OR LEN(ARRAY$(X))<1 THEN 200
230   FOR G=1 TO 15-LEN(ARRAY$(X))
240     LET ARRAY$(X)=ARRAY$(X)&" "
250   NEXT G
260 NEXT X
270 CALL SORT
280 CALL DUMP
290 DEF SORT
300   PRINT :PRINT "Sorting, please wait.":PRINT
310   FOR J=1 TO AMOUNT-1
320     FOR I=J+1 TO AMOUNT
330       LET L=AMOUNT+J-I+1
340       IF ARRAY$(J)>ARRAY$(L) THEN
350         LET M=J
360         LET TEMP$=ARRAY$(L)
370         LET ARRAY$(L)=ARRAY$(J)
380         LET ARRAY$(J)=TEMP$
390       END IF
400     NEXT I
410   NEXT J
420 END DEF
430 DEF DUMP
440   FOR X=1 TO AMOUNT
450     PRINT ARRAY$(X)
460   NEXT X
470 END DEF
```